TARGET = qwordcards
BUILD_PATH = ./build
QMAKE_CXXFLAGS_RELEASE = -O3
CONFIG += warn_on
win32:LIBS += -lws2_32
CONFIG(debug, debug|release) { 
    # Debug
    DEFINES += DEBUG
    DEFINES -= RELEASE
    CONFIG -= console
    DESTDIR = $${BUILD_PATH}/debug
    unix:TARGET = $$join(TARGET,,,_debug)
    else:TARGET = $$join(TARGET,,,d)
    unix:OBJECTS_DIR = $${BUILD_PATH}/debug/.obj/unix
    win32:OBJECTS_DIR = $${BUILD_PATH}/debug/.obj/win32
    mac:OBJECTS_DIR = $${BUILD_PATH}/debug/.obj/mac
    UI_DIR = $${BUILD_PATH}/debug/.ui
    MOC_DIR = $${BUILD_PATH}/debug/.moc
    RCC_DIR = $${BUILD_PATH}/debug/.rcc
}
else { 
    # Release
    DEFINES -= DEBUG
    DEFINES += RELEASE
    DESTDIR = $${BUILD_PATH}/release
    unix:OBJECTS_DIR = $${BUILD_PATH}/release/.obj/unix
    win32:OBJECTS_DIR = $${BUILD_PATH}/release/.obj/win32
    mac:OBJECTS_DIR = $${BUILD_PATH}/release/.obj/mac
    UI_DIR = $${BUILD_PATH}/release/.ui
    MOC_DIR = $${BUILD_PATH}/release/.moc
    RCC_DIR = $${BUILD_PATH}/release/.rcc
}
HEADERS += src/wgtsettings.h \
    src/textlabel.h \
    src/preferences.h \
    src/mainwindow.h \
    src/dictionaryloadthread.h \
    src/word.h \
    src/dictionary.h \
    src/dictionariesmanager.h \
    src/wordindex.h \
    src/dictziploader.h \
    src/wordslist.h \
    src/dictionarymediator.h \
    src/mediatorsmanager.h
SOURCES += src/word.cpp \
    src/wgtsettings.cpp \
    src/textlabel.cpp \
    src/preferences.cpp \
    src/mainwindow.cpp \
    src/main.cpp \
    src/dictionaryloadthread.cpp \
    src/dictionary.cpp \
    src/dictionariesmanager.cpp \
    src/wordindex.cpp \
    src/dictziploader.cpp \
    src/wordslist.cpp \
    src/dictionarymediator.cpp \
    src/mediatorsmanager.cpp
RESOURCES += src/resources.qrc
